PARAMETERS
A         intercept of supply on the P axis (MC at Q = 0)
B         change in MC in response to Q – this is dP over dQ
C         intercept of demand on the Q axis (demand at P = 0)
D         response of demand to changes in price – dQ over dP
TAX       a tax rate used later for experiments;

A = 2;
C = 6;
B = 1;
D = -1;

NONNEGATIVE VARIABLES
P         price of good X
X         quantity of good X;

EQUATIONS
SUPPLY    supply relationship (marginal cost ge price)
DEMAND    quantity demanded as a function of price;

SUPPLY..   A + B*X  =G=  P;
DEMAND..  X  =G=  C + D*P;

MODEL  EQUIL  /SUPPLY.X, DEMAND.P/;
OPTION  MCP = PATH;
SOLVE  EQUIL USING MCP;

* counter factual 1: shift the supply curve (marginal cost) up/left

A = 7;
SOLVE  EQUIL USING MCP;

* country factual 2: shift the supply curve (marginal cost) down/right

A = -7;
SOLVE  EQUIL USING MCP;

* exercise 1: extract economic information from the solution

PARAMETERS
  CONSPRICE    consumer price
  PRODPRICE    producer price (equal to marginal cost)
  TAXREV       tax revenue (note tax base is producer price)
CONSURP1 consumer surplus with no tax
CONSURP2 consumer surplus with 25% tax
PROSURP1 producer "surplus" with no tax
PROSURP2 producer "surplus" with 25% tax
DWL net loss from the tax;

EQUATIONS
SUPPLY2;

SUPPLY2.. (A + B*X)*(1+TAX) =G= P;

MODEL EQUIL2 /SUPPLY2.X, DEMAND.P/;

A = 2;
TAX = 0;
SOLVE EQUIL2 USING MCP;

CONSURP1 = (-C/D - P.L)*X.L/2;
PROSURP1 = (P.L/(1+TAX) - A)*X.L/2;

TAX = 0.25;
SOLVE EQUIL2 USING MCP;

CONSURP2 = (-C/D - P.L)*X.L/2;
PROSURP2 = (P.L/(1+TAX) - A)*X.L/2;
CONSPRICE = P.L;
PRODPRICE = P.L/(1+TAX);
TAXREV = PRODPRICE*TAX*X.L;
**DISPLAY** CONSPRICE, PRODPRICE, TAXREV;

DWL = CONSURP1 + PROSURP1 - (CONSURP2 + PROSURP2 + TAXREV);
**DISPLAY** CONSURP1, PROSURP1, CONSURP2, PROSURP2, TAXREV, DWL;

*exercise 2, mismatch the complementary variables*

TAX = 0;

**MODEL** EQUIL3 /SUPPLY.P, DEMAND.X/;

**SOLVE** EQUIL3 USING MCP;

P.L = 0;
X.L = 6;

A = 7;
**SOLVE** EQUIL3 USING MCP;

A = -7;
**SOLVE** EQUIL3 USING MCP;